УДК 519.6

Геометрические приложения двойного интеграла: визуализация в системе Maple

Прохорова Наталья Юрьевна
Приамурский государственный университет имени Шолом-Алейхема, студент

Сизинцева Анастасия Александровна
Приамурский государственный университет имени Шолом-Алейхема, студент

Эйрих Надежда Владимировна
Приамурский государственный университет имени Шолом-Алейхема
к.ф.-м.н., доцент

Аннотация
Графические возможности системы Maple использованы для изображения плоских областей и пространственных тел, ограниченных заданными кривыми и поверхностями. Описан алгоритм построения изображения цилиндрического бруса, объем которого задает двойной интеграл от функции *f(x,y)* по области Ω.
Ключевые слова: двойной интеграл, цилиндрический брус

Geometric Applications of Double Integral: Imaging in Maple System

Prokhorova Natalya Yurievna
Sholom-Aleichem Priamursky State University, student

Sizinceva Anastasiya Alexandrovna
Sholom-Aleichem Priamursky State University, student

Eyrikh Nadezhda Vladimirovna
Sholom-Aleichem Priamursky State University
PhD in Mathematics, Associate Professor

Abstract
Graphic features Maple system used to image areas of flat and three-dimensional bodies bounded by given curves and surfaces. The algorithm of construction of the cylindrical timber of the image, which determines the amount of the double integral of a function *f(x,y)* on the field Ω.
Keywords: double integral, cylindrical timber

 ISSN 2414-4487

Из курса математического анализа известно, что если подынтегральная функция $f(x,y) > 0$, $(x,y) \in \Omega$, то двойной интеграл $\iint\limits_{\Omega} f(x,y)\,dxdy$ равен объему цилиндрического бруса, нижним основанием которого является плоская область Ω, верхним основанием служит график функции $f(x,y)$, а боковая поверхность образована прямыми, параллельными оси Oz. Если же подынтегральная функция $f(x,y) \equiv 1$, $(x,y) \in \Omega$, то двойной интеграл $\iint\limits_{\Omega} dxdy$ дает площадь плоской области Ω [3].

При решении задач на геометрические приложения двойного интеграла сложности возникают именно на этапе построения заданных тел и областей. Графические возможности математического пакета Maple позволяют визуализировать плоские области и пространственные тела, площади и объемы которых требуется найти с помощью двойного интеграла [2, 4, 5]. Продемонстрируем это на конкретных примерах.

Пример 1. Найти двойным интегрированием площадь области, ограниченной линиями $y^2 = x$ и $y = x^2$ (рис. 1).

```
> with(plots) : with(plottools) :
> #Область Ω ограничена линиями y=x^2 и y = √x
> d := inequal({y - x^2 ≥ 0, y - √x ≤ 0}, x=-1..2, y=-1..2, color = "BlueViolet") :
  l := line([0, 1], [1, 1], linestyle = 2), line([1, 0], [1, 1], linestyle = 2) :
  t := textplot([1.7, 2,'y = x^2','font'= ["times", "roman", 17]]), textplot([1.7, 1.1,'y = √x','font'
      = ["times", "roman", 17]]), textplot([0.5, 0.5,'Ω','font'= ["times", "roman", 17]]) :
  display(d, l, t);
```



Рисунок 1 – Область, ограниченная линиями $y^2 = x$ и $y = x^2$

 ISSN 2414-4487

Дадим описание используемых команд:

- ***with(plots):with(plottools):*** – подключение дополнительных пакетов графических расширений;
- ***inequal*** – создает графический объект ***d*** – область, ограниченную заданными неравенствами и заливает эту область указанным цветом;
- ***line*** – создает графический объект ***l*** – пунктирные линии (опция ***linestyle = 2***), демонстрирующие координаты точки пересечения заданных кривых;
- ***textplot*** – создает графический объект ***t*** – текст, содержащий уравнения изображенных кривых (опция ***font*** позволяет управлять типом и размером шрифта надписи);
- ***display*** – изображает созданные графические объекты в одной плоскости.

Пример 2. Найти двойным интегрированием площадь области, ограниченной прямыми $y = x$, $y = 2x$, $x = 1$ и $x = 2$ (рис. 2).

```
> d := inequal({y − x ≥ 0, y − 2·x ≤ 0, x ≥ 1, x ≤ 2}, x = 0..3, y = 0..5, color
      = "DarkMagenta", scaling = constrained) :
  l := line([0, 1], [1, 1], linestyle = 2), line([0, 2], [2, 2], linestyle = 2), line([0, 4], [2, 4],
      linestyle = 2) :
  t := textplot([2.5, 4.3,'y = 2x','font'= ["times", "roman", 17]]), textplot([2.5, 2.2,'y = x','font'
      = ["times", "roman", 17]]), textplot([1.5, 2.2,'Ω','font'= ["times", "roman", 17]]) :
  display(d, l, t);
```



Рисунок 2 – Область, ограниченная прямыми $y = x$, $y = 2x$, $x = 1$ и $x = 2$

Пример 3. Найти двойным интегрированием площадь области, ограниченной линиями $x^2 + y^2 - 2ax = 0$ и $x^2 + y^2 - ax = 0$ (рис. 3).

```
> d := inequal({x^2 + y^2 - 2·a·x ≤ 0, x^2 + y^2 - a·x ≥ 0}, x=-1..3, y=-1..1.5, color
      = "Coral") :
  t := textplot([1.5, 1.2,'x^2 + y^2 - 2 ax = 0','font'= ["times", "roman", "bold", 14]]),
      textplot([1.5, 0.5,'x^2 + y^2 - ax = 0','font'= ["times", "roman", "bold", 14]]) :
  display(d, t, scaling = constrained);
```



Рисунок 3 – Область, ограниченная линиями $x^2 + y^2 - 2ax = 0$ и $x^2 + y^2 - ax = 0$, при $a = 1$

Пример 4. Найти двойным интегрированием объем тела, ограниченного заданными поверхностями: параболоидом вращения $z = x^2 + y^2$, координатными плоскостями и плоскостью $x + y = 1$ [1].

Для того чтобы «увидеть» тело необходимо понять:

1) что является подынтегральной функцией $f(x, y)$,
2) что представляет собой область интегрирования, т.е. на какой плоской области задана подынтегральная функция.

Для ответов на эти вопросы вначале изображаем заданные поверхности. При этом всякий раз подбираем оптимальные значения границ изменения координат, позволяющие получить более информативное изображение (рис. 4). Используемая в данном случае команда ***implicitplot3d*** позволяет строить в пространстве графики неявных функций, для одновременного вывода графиков нескольких неявных функций их уравнения перечисляют через запятую и заключают в квадратные скобки. Опция ***scaling*** управляет масштабом координатных осей (значение опции ***constrained*** делает масштаб по всем осям одинаковым). Опция ***color*** задает цвет поверхности (названия всех цветов широкой палитры, имеющейся в

Maple, можно найти через Maple Help, задав в строке поиска colorname). Опция ***axes*** управляет видом координатных осей (значение normal дает стандартное изображение координатных осей, пересекающихся в начале координат).

```
> implicitplot3d([z = x^2 + y^2, x + y = 1], x = -1.7..1.7, y = -1.7..1.7, z = 0..2.5, color = [blue,
    green], scaling = constrained, axes = normal)
```



Рисунок 4 – Параболоид вращения $z = x^2 + y^2$ и плоскость $x + y = 1$

Из рисунка 4 видно, что заданное тело сверху ограничено параболоидом вращения $z = x^2 + y^2$ (синий цвет), а в нижнем основании лежит треугольник, образованный прямыми $x + y = 1$, $x = 0$ и $y = 0$. Изображаем нижнее и верхнее основания цилиндрического бруса (рис. 5).

```
> d := plot3d(0, x = 0..1 - y, y = 0..1, color = "BlueViolet") :
  f := plot3d(x^2 + y^2, x = 0..1 - y, y = 0..1, color = blue, scaling = constrained, axes
      = boxed) :
  display(d, f);
```



Рисунок 5 – Верхнее и нижнее основания цилиндрического бруса

Для получения боковой поверхности создаем семейство линий, параллельных оси аппликат, соединяющих соответствующие точки нижнего и верхнего оснований (рис. 6).

```
> N := 200 :
  for i from 1 to N do x[i] := i/N; y2[i] := i/N; y3[i] := (1 - x[i]); z1[i] := (x[i])^2; z2[i]
     := (y2[i])^2; z3[i] := (x[i])^2 + (y3[i])^2;
                    l1[i] := line([x[i], 0, 0], [x[i], 0, z1[i]], linestyle = 1 , thickness = 2);
                    l2[i] := line([0, y2[i], 0], [0, y2[i], z2[i]], linestyle = 1, thickness = 2);
                    l3[i] := line([x[i], y3[i], 0], [x[i], y3[i], z3[i]], linestyle = 1, thickness = 2) od:
  bp1 := seq(l1[i], i = 1 ..N) :
  bp2 := seq(l2[i], i = 1 ..N) :
  bp3 := seq(l3[i], i = 1 ..N) :
  display(d, f, bp1, bp2, bp3, view = [0 ..1, 0 ..1, 0 ..1])
```



Рисунок 6 – Цилиндрический брус, ограниченный параболоидом вращения $z = x^2 + y^2$, координатными плоскостями и плоскостью $x + y = 1$

Пример 5. Найти двойным интегрированием объем тела, ограниченного цилиндрами $x^2 + y^2 = R^2$ и $x^2 + z^2 = R^2$ [1].

Задаем какое-нибудь значение R и изображаем в одной системе координат эти два круговых цилиндра (рис. 7). Увидев пересечение цилиндров, создаем три графических объекта: ***d*** – круг $x^2 + y^2 \le R^2$ (область интегрирования), ***f1*** – часть цилиндра $z = \sqrt{R^2 - x^2}$ (верхнее основание), ***f2*** – часть цилиндра $z = -\sqrt{R^2 - x^2}$ (нижнее основание). Изображаем полученные объекты в одной системе координат (рис. 8).

```
> implicitplot3d([x^2 + y^2 = R^2, x^2 + z^2 = R^2], x = -R..R, y = -R..R, z = -R..R, color = [blue, green],
    scaling = constrained, axes = boxed)
```



Рисунок 7 – Цилиндры $x^2 + y^2 = R^2$ и $x^2 + z^2 = R^2$ при $R = 2$

```
> d := plot3d(0, x = -R..R, y = -sqrt(R^2 - x^2)..sqrt(R^2 - x^2), color = "BlueViolet") :
  f1 := plot3d(sqrt(R^2 - x^2), x = -R..R, y = -sqrt(R^2 - x^2)..sqrt(R^2 - x^2), color = blue, scaling = constrained,
      axes = boxed) :
  f2 := plot3d(-sqrt(R^2 - x^2), x = -R..R, y = -sqrt(R^2 - x^2)..sqrt(R^2 - x^2), color = blue, scaling
      = constrained, axes = boxed) :
  display(d, f1, f2);
```



Рисунок 8 – Верхнее и нижнее основания цилиндрического бруса

Аналогично предыдущему примеру создаем боковую поверхность (рис. 9) и изображаем искомый цилиндрический брус (рис. 10). Следует также отметить, что созданные 3D-объекты можно вращать, тем самым улучшая зрительное восприятие объемной фигуры (рис. 10,a и рис. 10,b).

```
> N := 100 :
  for i from 1 to N do x[i] := -R + i·2·R/N; y1[i] := sqrt(R^2 - (x[i])^2); y2[i] := -sqrt(R^2 - (x[i])^2);
     z1[i] := sqrt(R^2 - (x[i])^2); z2[i] := -sqrt(R^2 - (x[i])^2);
                     l1[i] := line([x[i], y1[i], z2[i]], [x[i], y1[i], z1[i]], linestyle = 1 , thickness
     = 2);
                     l2[i] := line([x[i], y2[i], z2[i]], [x[i], y2[i], z1[i]], linestyle = 1, thickness = 2)
     od:
  bp1 := seq(l1[i], i = 1 ..N) :
  bp2 := seq(l2[i], i = 1 ..N) :
  display(d, f1, f2, bp1, bp2, view = [-R ..R, -R ..R, -R ..R])
```

Рисунок 9 – Создание семейства линий, образующих боковую поверхность цилиндрического бруса



a) b)

Рисунок 10 – Цилиндрический брус, ограниченный цилиндрами $x^2 + y^2 = R^2$ и $x^2 + z^2 = R^2$

Уникальные возможности системы Maple по созданию двух- и трехмерных компьютерных моделей являются ее несомненным достоинством. Визуализированные с помощью современных компьютерных технологий математические знания позволяют более продуктивно вести учебный процесс, что ведет к повышению качества подготовки.

Библиографический список

1. Берман Г.Н. Сборник задач по курсу математического анализа. М.: Физматлит, 1985. 384 с.
2. Дьяконов В.П. Maple 9.5/10 в математике, физике и образовании. М.:СОЛОН-Пресс, 2006. 720 с.
3. Кудрявцев Л.Д. Краткий курс математического анализа. Т.2. Дифференциальное и интегральное исчисления функций многих переменных. Гармонический анализ. М.: Физматлит, 2005. 424 с.
4. Сабадаш Т.Л., Эйрих Н.В. Геометрический смысл производной: визуализация в Maple // Постулат. 2015. № 1 (1). С. 15.
5. Сизинцева А.А., Эйрих Н.В. Анимация в системе Maple циклоидальных кривых // Постулат. 2015. № 2 (2). С. 1.